CERN-SL Note / 97-41 (AP)

Lifetime Reduction by a Single Resonance

John M. Jowett

3 May 1997

1 Introduction

I present a naive model which attempts to show how a large isolated resonance can lead to enhanced beam tails and lifetime reduction. This goes some way towards explaining the measurements of "dynamic aperture" by blowing up the beam on the (108,90) LEP optics in November 1996. It was originally outlined within my presentation at the LEP Performance Workshop in January [1].

This model was concocted for illustrative purposes only. It is by no means intended as a serious theory of the effects and I am aware that can be criticised on several grounds. It is one-dimensional. It does not attempt to describe particle dynamics inside the resonance. In reality the diffusive effects involve couplings to the other two modes of oscillation. The time-dependence and quantum lifetime are treated in the usual, sloppy, quasi-static way, leading to a distribution function which is not properly normalised. The true physics can be better studied by particle tracking. However I think this analytical model is useful as a quick route to qualitative understanding of how non-gaussian tails can arise and how the beam lifetime can be reduced.

This note contains the details that I did not give in my talk in Chamonix and that I gave only partially in the writeup. For the basic physics, I follow the notations and general methods given in [2].

This note is written in the form of a Mathematica 3.0 notebook [3] and is printed with all the "working" visible (no pencils were used in this production ...). Notations are a hybrid between conventional, often ambiguous, mathematical usage (generated on occasion by the function `TraditionalForm`) and the more precise forms imposed by Mathematica. I use features of the Mathematica language freely without explanation but I think this note can still be followed by someone who is unfamiliar with it. Input expressions are given in **`bold type`** and are generally printed in a box, together with the results of their evaluation by the Mathematica kernel. It is worth knowing that square brackets are used to denote the arguments of functions, as in `Sin[x]`, while round brackets are reserved to indicate grouping. Occasionally I use the alternative "postfix" notation // for functions applied as "wrappers" to an expression, e.g.,

```
Integrate[f[θ] / (I + Sin[θ - ξ]), {θ, 0, Pi}] // TraditionalForm
```

$$\int_0^{\pi} \frac{f(\theta)}{\sin(\theta-\xi)+i}\, d\theta$$

It is better to read this notebook interactively. It is available from the World-Wide Web page:

`http://wwwslap.cern.ch/jowett/JMJdocs.html`

2 Fokker-Planck Equation

2.1 Drift and diffusion terms

In horizontal betatron phase space, the distribution F[Ix] in the action variable `Ix` provides a reduced description of the beam in which terms involving the phase have been averaged out. The evolution of this distribution is governed by the time-dependent Fokker-Planck equation $\tau_x \partial_t F_0[Ix]$ = rhsFPE, where τ_x is the radiation damping time and the right-hand side is

$$\text{rhsFPE} = -\partial_{Ix}(\text{DI}[Ix]\,F[Ix]) + \frac{1}{2}\partial_{\{Ix,2\}}(\text{QIsq}[Ix]\,F[Ix])$$

$$2\,F[Ix] + 2\,(Ix - \epsilon_x)\,F'[Ix] + \frac{1}{2}\,(8\,\epsilon_x\,F'[Ix] + 4\,Ix\,\epsilon_x\,F''[Ix])$$

In the simplest case of linear betatron motion with linear radiation damping and quantum excitation, the drift and diffusion function are given by:

```
DI[Ix_] := -2 (Ix - ϵx);
QIsq[Ix_] := 4 ϵx Ix
```

and the right-hand side of the Fokker-Planck equation is simply

```
rhsFPE = Simplify[rhsFPE]
```

$$2\,(F[Ix] + (Ix + \epsilon_x)\,F'[Ix] + Ix\,\epsilon_x\,F''[Ix])$$

Here the parameter ϵ_x is just a measure of the strength of quantum-excitation, derived from first principles in the lectures referred to above.

In this note, we do not attempt to construct the drift and diffusion terms as modified by the resonance. This can be done by straightforward application of the methods of [2] in the isolated resonance approximation.

2.2 Form of Equilibrium Solution

2.2.1 A direct approach (that doesn't work ...)

Directly solve the above differential equation for an equilibrium distribution

```
DSolve[rhsFPE == 0, F, Ix] // PowerExpand
```

$$\left\{\left\{F \to \left(E^{-\frac{\text{Log}[\#1]}{2}-\frac{1}{2}\#1\,\epsilon_x^{-\frac{2}{2}}-\frac{\#1}{2\,\epsilon_x}}\left(C[2]\;\text{Hypergeometric1F1}\left[\frac{1}{2}-\frac{\epsilon_x^{-\frac{2(-1)}{2}}}{2\,\epsilon_x},1,\#1\,\epsilon_x^{-\frac{2}{2}}\right]+C[1]\;\text{HypergeometricU}\left[\frac{1}{2}-\frac{\epsilon_x^{-\frac{2(-1)}{2}}}{2\,\epsilon_x},1,\#1\,\epsilon_x^{-\frac{2}{2}}\right]\right)\sqrt{\#1}\,\&\right)\right\}\right\}$$

2.2.2 Alternative (that does work ...)

The direct approach gives a complicated representation of something which is really fairly simple. Let's do the first integral by hand.

$$\text{FPEint} = -\text{DI}[\text{Ix}]\,F[\text{Ix}] + \frac{1}{2}\,\partial_{\text{Ix}}\,(\text{QIsq}[\text{Ix}]\,F[\text{Ix}]) == C[0]$$

$$2\,F[\text{Ix}]\,(\text{Ix}-\epsilon_x)+\frac{1}{2}\,(4\,F[\text{Ix}]\,\epsilon_x+4\,\text{Ix}\,\epsilon_x\,F'[\text{Ix}]) == C[0]$$

Here is the basic solution of the one-dimensional equation with two undetermined constants that are to be use d to match the boundary conditions.

```
F0basic = First[DSolve[FPEint, F, Ix]]
```

$$\left\{F \to \left(E^{-\frac{\#1}{\epsilon_x}}\,C[1]+\frac{E^{-\frac{\#1}{\epsilon_x}}\,C[0]\;\text{ExpIntegralEi}\left[\frac{\#1}{\epsilon_x}\right]}{2\,\epsilon_x}\,\&\right)\right\}$$

```
F0form = F[Ix] /. F0basic
```

$$E^{-\frac{\text{Ix}}{\epsilon_x}}\,C[1]+\frac{E^{-\frac{\text{Ix}}{\epsilon_x}}\,C[0]\;\text{ExpIntegralEi}\left[\frac{\text{Ix}}{\epsilon_x}\right]}{2\,\epsilon_x}$$

In traditional notation, this is

```
F0form // TraditionalForm
```

$$e^{-\frac{\text{Ix}}{\epsilon_x}}\,c_1+\frac{e^{-\frac{\text{Ix}}{\epsilon_x}}\,c_0\,\text{Ei}\left(\frac{\text{Ix}}{\epsilon_x}\right)}{2\,\epsilon_x}$$

2.3 Properties of basic solutions

2.3.1 Exponential integral

```
Plot[ExpIntegralEi[x], {x, 0, 1}]
```




```
- Graphics -
```

Some derivatives of the exponential integral:

```
TraditionalForm[
 TableForm[Table[{n, ∂{x,n} ExpIntegralEi[x]}, {n, 0, 4}]]]
```

0	$\mathrm{Ei}(x)$
1	$\frac{e^x}{x}$
2	$\frac{e^x}{x} - \frac{e^x}{x^2}$
3	$\frac{e^x}{x} - \frac{2e^x}{x^2} + \frac{2e^x}{x^3}$
4	$\frac{e^x}{x} - \frac{3e^x}{x^2} + \frac{6e^x}{x^3} - \frac{6e^x}{x^4}$

2.3.2 Two basic solutions

Extract and name the two basic solutions

```
F1[x_] = Coefficient[F[x] /. F0basic /. ϵx → 1, C[1]];
F2[x_] = Coefficient[F[x] /. F0basic /. ϵx → 1, C[0]];
```

```
{F1[x], F2[x]} // TraditionalForm
```

$\left\{e^{-x}, \frac{1}{2} e^{-x} \mathrm{Ei}(x)\right\}$

```
Plot[{F1[x], F2[x]}, {x, 0, 10}]
```




```
- Graphics -
```

The second solution is excluded in the normal situation where we have to keep the density finite at the origin. However both vanish at infinity. To see this, it is not enough to evaluate

```
{F1[∞], F2[∞]}
∞::indet : Indeterminate expression 0 ∞ encountered.
{0, Indeterminate}
```

Although Ei(x) has an essential singularity at $x = \infty$, the `Series` function can derive a useful asymptotic expansion around it.

```
asymEi[x_] = Series[ExpIntegralEi[x], {x, ∞, 5}]
```

$$E^x \left(\frac{1}{x} + \left(\frac{1}{x}\right)^2 + 2\left(\frac{1}{x}\right)^3 + 6\left(\frac{1}{x}\right)^4 + 24\left(\frac{1}{x}\right)^5 + O\left[\frac{1}{x}\right]^6\right)$$

From this we can construct an approximation to `F2`:

```
asymF2[x_] = Normal[e^-x/2 asymEi[x]]

Series::esss : Essential singularity encountered in Exp[\(\*InterpretationBox[
     1\/Series`n + O[Series`n]^7, SeriesData[Series`n, 0, List[1], -1, 7, 1]]\)].

Series::esss : Essential singularity encountered
   in Exp[\(\*InterpretationBox[\(-\(1\/Series`n\)\) + O[Series`n]^7,
     SeriesData[Series`n, 0, List[-1], -1, 7, 1]]\)].

Series::esss : Essential singularity encountered in Exp[\(\*InterpretationBox[
     1\/Series`n + O[Series`n]^7, SeriesData[Series`n, 0, List[1], -1, 7, 1]]\)].

General::stop :
 Further output of Series::esss will be suppressed during this calculation.
```

$$\frac{12}{x^5} + \frac{3}{x^4} + \frac{1}{x^3} + \frac{1}{2\,x^2} + \frac{1}{2\,x}$$

The following logarithmic plot of the fractional error shows that the first few terms of the asymptotic series do indeed provide an excellent approximation to the function:

```
Plot[Log[Abs[asymF2[x]/F2[x] - 1]], {x, 5, 1000}, PlotRange -> All]
```



- Graphics -

The asymptotic form confirms that the second function `F2` "vanishes at infinity" (although it is not integrable).

3 Matching Boundary Conditions

We can use the basic solution with two undetermined constants in different ranges of the action `Ix` in which the dynamics will be assumed to be linear. Different treatment will be required in regions where the dynamics is nonlinear.

We can determine the constants `C[0]` and `C[1]` in each region by matching the boundary conditions.

3.1 Linear Neigbourhood of the Origin

The distribution function must be finite at the origin. Call its value `F00`. This leads to a rule for determining one of the constants.

```
finiteOrigin = First[Solve[F[0] == F00 /. F0basic, C[0]]]
{C[0] → 0}
```

Here is a temporary version of the solution in the linear region close to the origin:

```
Fat[Ix_] = F[Ix] /. F0basic /. finiteOrigin
```

$E^{-\frac{Ix}{\epsilon_x}}\,C[1]$

Eliminate the other constant by using a value at the origin:

```
valueOrigin = First[Solve[Fat[0] == F00, C[1]]]
{C[1] → F00}
```

```
Fa[Ix_] := F[Ix] /. F0basic /. finiteOrigin /. valueOrigin;
Fa[Ix]
```

$E^{-\frac{Ix}{\epsilon_x}}\,F00$

In the case where the linear region extends to infinity, we can compute F00 by normalisation of the distribution:

$Z_0 = \int_0^\infty Fa[Ix]\,dIx$

$If\left[Re[\epsilon_x] > 0,\ F00\,\epsilon_x,\ \int_0^\infty E^{-\frac{Ix}{\epsilon_x}}\,F00\,dIx\right]$

We know that ϵ_x is a positive real quantity so let's make the appropriate assumption.

$Z_0 = Z_0$ /. $Re[\epsilon_x] > 0$ -> True

$F00\,\epsilon_x$

which leads to a rule and a new definition of `Fa`:

$normOrule = First[Solve[Z_0 == 1, F00]]$

$\left\{F00 \to \frac{1}{\epsilon_x}\right\}$

```
Fa[Ix_] := F[Ix] /. F0basic /. finiteOrigin /. valueOrigin /. norm0rule;
Fa[Ix]
```

$$\frac{E^{-\frac{Ix}{\epsilon_x}}}{\epsilon_x}$$

The density at the origin is then just

```
Limit[Fa[Ix], Ix -> 0]
```

$$\frac{1}{\epsilon_x}$$

3.2 Resonance-dominated region

Assume there is a region $I_{rmin} < I_x < I_{rmax}$ dominated by a single resonance. The model is based on the crude approximation that the density is constant in this region, reflecting the physical idea that particles are transported rapidly from the inner to the outer side of the resonance "island" and back again. In this region, the solution is therefore taken to be:

```
Fb[Ix_] = Fa[Ir_min]
```

$$\frac{E^{-\frac{Ir_{min}}{\epsilon_x}}}{\epsilon_x}$$

3.3 Linear region above resonance, Version A

Above the resonance layer, the dynamics is assumed to become linear once again. Hence the solution will be of the form given by the rule `F0basic` with constants to be determined by matching the boundary conditions at I_{rmax}.

Here is a first version, in which the distribution function is supposed continuous and its derivative is set, somewhat arbitrarily, to zero.

```
IrmaxMatchA = First[Solve[{(F[Ir_max] /. F0basic) == Fb[Ir_max],
(F'[Ir_max] /. F0basic) == 0}, {C[0], C[1]}]]
```

$$\left\{C[0] \to \frac{2\,E^{-\frac{Ir_{min}}{\epsilon_x}}\,Ir_{max}}{\epsilon_x},\; C[1] \to \frac{E^{-\frac{Ir_{min}}{\epsilon_x}}\left(-\text{ExpIntegralEi}\left[\frac{Ir_{max}}{\epsilon_x}\right] Ir_{max} + E^{\frac{Ir_{max}}{\epsilon_x}}\,\epsilon_x\right)}{\epsilon_x^2}\right\}$$

```
FcA[Ix_] = Simplify[F[Ix] /. F0basic /. IrmaxMatchA]
```

$$\frac{1}{\epsilon_x^2}\left(E^{-\frac{Ix+Ir_{min}}{\epsilon_x}}\left(\left(\text{ExpIntegralEi}\left[\frac{Ix}{\epsilon_x}\right] - \text{ExpIntegralEi}\left[\frac{Ir_{max}}{\epsilon_x}\right]\right) Ir_{max} + E^{\frac{Ir_{max}}{\epsilon_x}}\,\epsilon_x\right)\right)$$

Check that we really satisfied our boundary conditions:

```
Simplify[{FcA[Ir_max], FcA'[Ir_max]}]
```

$$\left\{\frac{E^{-\frac{Ir_{min}}{\epsilon_x}}}{\epsilon_x},\ 0\right\}$$

3.4 Linear region above resonance, Version B

Instead of requiring continuity of the derivative of the distribution, we can, instead, arbitrarily reduce the value of `C[0]` by the following trick:

```
F'[Ir_max] /. F0basic
```

$$-\frac{E^{-\frac{Ir_{max}}{\epsilon_x}}\,C[0]\ \text{ExpIntegralEi}\left[\frac{Ir_{max}}{\epsilon_x}\right]}{2\,\epsilon_x^2}-\frac{E^{-\frac{Ir_{max}}{\epsilon_x}}\,C[1]}{\epsilon_x}+\frac{C[0]}{2\,Ir_{max}\,\epsilon_x}$$

```
IrmaxMatchB = First[Solve[{(F[Ir_max] /. F0basic) == Fb[Ir_max],
(F'[Ir_max] /. F0basic) == Fa'[Ir_max]}
, {C[0], C[1]}]]
```

$$\left\{C[0]\to-\frac{2\,E^{-\frac{Ir_{max}}{\epsilon_x}-\frac{Ir_{min}}{\epsilon_x}}\left(-E^{\frac{Ir_{max}}{\epsilon_x}}+E^{\frac{Ir_{min}}{\epsilon_x}}\right)Ir_{max}}{\epsilon_x},\right.$$
$$C[1]\to\frac{1}{\epsilon_x^2}\left(E^{-\frac{Ir_{max}}{\epsilon_x}-\frac{Ir_{min}}{\epsilon_x}}\left(-E^{\frac{Ir_{max}}{\epsilon_x}}\ \text{ExpIntegralEi}\left[\frac{Ir_{max}}{\epsilon_x}\right]Ir_{max}+\right.\right.$$
$$\left.\left.\left.E^{\frac{Ir_{min}}{\epsilon_x}}\ \text{ExpIntegralEi}\left[\frac{Ir_{max}}{\epsilon_x}\right]Ir_{max}+E^{\frac{2\,Ir_{max}}{\epsilon_x}}\,\epsilon_x\right)\right)\right\}$$

```
FcB[Ix_] = Simplify[F[Ix] /. F0basic /. IrmaxMatchB]
```

$$-\frac{1}{\epsilon_x^2}\left(E^{-\frac{Ix+Ir_{max}+Ir_{min}}{\epsilon_x}}\right.$$
$$\left(-\left(E^{\frac{Ir_{max}}{\epsilon_x}}-E^{\frac{Ir_{min}}{\epsilon_x}}\right)\left(\text{ExpIntegralEi}\left[\frac{Ix}{\epsilon_x}\right]-\text{ExpIntegralEi}\left[\frac{Ir_{max}}{\epsilon_x}\right]\right)Ir_{max}-\right.$$
$$\left.\left.E^{\frac{2\,Ir_{max}}{\epsilon_x}}\,\epsilon_x\right)\right)$$

Check that we really satisfied our boundary conditions:

```
Simplify[{FcB[Ir_max], FcB'[Ir_max]} ]
```

$$\left\{\frac{E^{-\frac{Ir_{min}}{\epsilon_x}}}{\epsilon_x},\ -\frac{E^{-\frac{Ir_{max}}{\epsilon_x}}}{\epsilon_x^2}\right\}$$

This solution goes to zero at infinity:

```
Simplify[FcB[∞ ϵ□]]
```

$$-\frac{1}{\epsilon_x^2}\left(E^{-\frac{Ir_{max}+Ir_{min}+\infty\,\epsilon_\square}{\epsilon_x}}\left(\left(E^{\frac{Ir_{max}}{\epsilon_x}}-E^{\frac{Ir_{min}}{\epsilon_x}}\right)\left(\text{ExpIntegralEi}\left[\frac{Ir_{max}}{\epsilon_x}\right]-\text{ExpIntegralEi}\left[\frac{\infty\,\epsilon_\square}{\epsilon_x}\right]\right)Ir_{max}-E^{\frac{2\,Ir_{max}}{\epsilon_x}}\,\epsilon_x\right)\right)$$

3.5 Numerical Example

This set of rules gives some typical numerical values, corresponding to a case where the resonance is at an amplitude corresponding to about 6σ of the beam distribution.

```
testvals = {ϵx → 1, Irmin → 16, Irmax → 20, τx → 1, F00 → 1}
{ϵx → 1, Irmin → 16, Irmax → 20, τx → 1, F00 → 1}
```

We can define the function globally with appropriate conditionals:

```
F0g[Ix_] := Which[
Ix < Irmin, Fa[Ix],
Ix < Irmax, Fa[Irmin],
Ix ≥ Irmax, FcB[Ix]
]
```

Some numerical values

```
N[{F0g[0.], F0g[.6], F0g[3], F0g[4], F0g[10]} //. testvals]
{1., 0.548812, 0.0497871, 0.0183156, 0.0000453999}
```

and a numerical check of the normalisation

```
NIntegrate[N[F0g[Ix] //. testvals], {Ix, 0, 50}]
1.
```

Plot the logarithm of the distribution function obtained by solving the Fokker-Planck equation with the above parameters and one choice of boundary condition. (Besides the computed distribution function, some additional graphical elements are added here.) The horizontal scale is x/σ_x.

```
Plot[Log[F0g[nsig^2/2] //. testvals], {nsig, 0, 10},
PlotRange → All,
AxesLabel -> {"x/σx", "F"},
Prolog -> {
{nsmin, nsmax, lFr} = {Sqrt[2 Irmin], Sqrt[2 Irmax]
, Log[F0g[Irmax]]} //. testvals;
GrayLevel[.9],
Rectangle[{nsmin, .1 lFr}, {nsmax, lFr}],
GrayLevel[0],
Text["Resonance layer", {(nsmin + nsmax) / 2, lFr / 2}, {0, 0}, {0, 1}]
}]
```




```
- Graphics -
```

A similar plot of the diffusive flux shows that it is continuous, as it must be for particle conservation.

$$\text{Plot}\left[\text{Log}\left[\frac{2\,Ix\,F0g[Ix]}{\tau_x} \;/\!/.\; \text{testvals}\right],\ \{Ix,\ 0,\ 40\}\right]$$



- Graphics -

3.6 Normalisation of the distribution

In principle, we can get the value of F00 by integrating the distribution piecewise.

$$Z = \int_0^{Ir_{max}} Fa[Ix]\,dIx + Fa[Ir_{min}]\,(Ir_{max} - Ir_{min}) + \int_{Ir_{max}}^{\infty} FcB[Ix]\,dIx$$

$$1 - E^{-\frac{Ir_{max}}{\epsilon_x}} - \frac{1}{\epsilon_x^2}\left(\int_{Ir_{max}}^{\infty} E^{-\frac{Ix+Ir_{max}+Ir_{min}}{\epsilon_x}}\left(-\left(E^{\frac{Ir_{max}}{\epsilon_x}} - E^{\frac{Ir_{min}}{\epsilon_x}}\right)\left(\text{ExpIntegralEi}\left[\frac{Ix}{\epsilon_x}\right] - \text{ExpIntegralEi}\left[\frac{Ir_{max}}{\epsilon_x}\right]\right) Ir_{max} - E^{\frac{2\,Ir_{max}}{\epsilon_x}}\,\epsilon_x\right) dIx\right) + \frac{E^{-\frac{Ir_{min}}{\epsilon_x}}\,(Ir_{max} - Ir_{min})}{\epsilon_x}$$

There is no analytic form for the final integral and, besides, it is logarithmically divergent, so the calculation does not yield a useful result. This is not a serious concern as we should not expect to be able to normalise a distribution which is slowly leaking away. A similar divergence arises when boundary conditions are properly imposed in any first-passage time problem. The distribution just has to be cut off at some finite amplitude.

In the numerical example above, we saw that the normalisation F00 -> $1/\epsilon_x$ works very well in a typical case.

4 Quantum Lifetime

The outward diffusive flux in the equilibrium distribution is balanced by the inwared damping flux. We can estimate it as

```
Jdiff[Ix_] = Simplify[2 Ix F0g[Ix] / τx /. normrule]
```

$$\frac{1}{\tau_x}\left(2\,\text{Ix}\,\text{Which}[\text{Ix} < \text{Ir}_{min}, \text{Fa}[\text{Ix}], \text{Ix} < \text{Ir}_{max}, \text{Fa}[\text{Ir}_{min}], \text{Ix} \geq \text{Ir}_{max}, \text{FcB}[\text{Ix}]]\right)$$

Then, according to the usual, somewhat sloppy, argument, its inverse is the quantum lifetime

```
tauq[Ix_] := Simplify[1 / Jdiff[Ix]]
```

To get the usual "Sands" formula for quantum lifetime, we just have to put the aperture limit in the linear region below the resonance. In this case the formula reduces to

```
tauq0[Ix_] = tauq[Ix] /. (Ix < Irmin -> True);
tauq0[Ix] // TraditionalForm
```

$$\frac{e^{\frac{\text{Ix}}{\epsilon_x}}\,\epsilon_x\,\tau_x}{2\,\text{Ix}}$$

The lifetime is almost constant through the resonance layer, consistent with the basic physical idea of the model.

```
tauq[Ix] /. {Ix < Irmin -> False, Ix < Irmax -> True} // TraditionalForm
```

$$\frac{e^{\frac{\text{Ir}_{min}}{\epsilon_x}}\,\epsilon_x\,\tau_x}{2\,\text{Ix}}$$

Finally, the expression for the quantum lifetime in the region above the resonance is

```
tauq[Ix] //. {Ix < Irmin -> False, Ix < Irmax -> False, Ix ≥ Irmax -> True} //
 TraditionalForm
```

$$-\frac{e^{\frac{\text{Ix}+\text{Ir}_{max}+\text{Ir}_{min}}{\epsilon_x}}\,\epsilon_x^2\,\tau_x}{2\,\text{Ix}\left(-\left(e^{\frac{\text{Ir}_{max}}{\epsilon_x}}-e^{\frac{\text{Ir}_{min}}{\epsilon_x}}\right)\left(\text{Ei}\left(\frac{\text{Ix}}{\epsilon_x}\right)-\text{Ei}\left(\frac{\text{Ir}_{max}}{\epsilon_x}\right)\right)\text{Ir}_{max}-e^{\frac{2\,\text{Ir}_{max}}{\epsilon_x}}\,\epsilon_x\right)}$$

To show the effect of the resonance, we can plot the logarithms of the two lifetimes as a function of the number of "sigmas". The dashed line shows the "Sands" formula.

```
Plot[{Log[tauq[n^2/2] //. testvals], Log[tauq0[n^2/2] //. testvals]}
    , {n, 4, 10}, PlotRange → {0, 40},
  PlotStyle -> {{}, Dashing[{0.1, 0.05}]}]
```




```
- Graphics -
```

Although I took Version B of the solution for the above examples, the results are very similar with Version A or almost any other way of matching the boundary conditions. The main exception is the case where the derivative of the distribution is made to have the same value above and below the resonance $F'(Ir_{min}) = F'(Ir_{max})$. Then the distribution in coordinate space is gaussian at large amplitudes but with a larger effective σ_x than in the core region. In this version of the model, the lifetime is reduced according to a "Sands" formula with appropriate parameters, i.e., less dramatically than in the example presented above.

5 Conclusions

Despite its obvious limitations and its agnosticism concerning the dynamics of the resonance itself, a simple diffusion model indicates that enhanced tails and reduced lifetime are generic features of a distribution function perturbed by a resonance structure. This helps in understanding the measurements of usable aperture by phase-space inflation and semi-quantitatively reproduces the non-gaussian tails of the transverse distribution found in the tail scans [4].

6 References

[1] J.M. Jowett, "Non-linear Resonances: Predictions, Effects and Measurements", Proc. 7th LEP Performance Workshop, Chamonix, January 1997, CERN-SL/97-06 (DI) or
`http://www.cern.ch/CERN/Divisions/SL/publications/chamx97/PAPERS/JJ3_3.PDF`

[2] J.M. Jowett, "Introductory Statistical Mechanics for Electron Storage Rings", US Particle Accelerator School, AIP Conference Proceedings, No. 153 (1985). This may still be available as SLAC-PUB 4033 (1986).

[3] S. Wolfram, The Mathematica Book, 3rd Edition, Cambridge University Press, 1996.

[4] I. Reichel, "Dynamic Aperture and Tail Scans", Proc. 7th LEP Performance Workshop, Chamonix, January 1997, CERN-SL/97-06 (DI) or
`http://www.cern.ch/CERN/Divisions/SL/publications/chamx97/PAPERS/IR3_1.PDF`